

情報処理2 第6回

# 十進BASIC (3) 繰り返しのある計算 (続き), グラフィックス入門

かつらだ まさし  
桂田 祐史

2013年5月22日

この授業用の WWW ページは <http://www.math.meiji.ac.jp/~mk/syori2-2013/>

## 1 連絡事項

- 前回、情報処理教室の Windows 環境から、dviout が消えた騒ぎもあって、予定していたよりも時間を取られましたが、dviout は復元してくれたそうです。今日はこれまで通りに使えるはずです。
- 上のトラブルもあるので、レポート課題 5A の提出は遅れても構いません、と言いましたが、実際には提出してくれた人が多かったです。良かった。
- レポート課題 5B<sup>1</sup> の締切は1週ずらしました。

## 2 気を取り直して前回の続き (レポート課題 5B に向けて)

<http://www.math.meiji.ac.jp/~mk/syori2-2013/jouhousyori2-2013-05/node5.html>

## 3 十進BASICでグラフィックス

数学のためにコンピューターに計算をさせる場合、

1. 数値計算 (Numerical computation)
2. 数式処理 (Symbolic computation)
3. グラフィックス (Graphics, 可視化 visualization)

が3本柱と言われます。今日は十進 BASIC で簡単なグラフィックスをやってみましょう。

<sup>1</sup><http://www.math.meiji.ac.jp/~mk/syori2-2013/jouhousyori2-2013-05/node6.html>

### 3.1 イントロ

汎用のプログラミング言語で、その規格の中にグラフィックスを含んでいるもの<sup>2</sup>は実は珍しく (Java と BASIC くらいしか思い浮かばない…)、十進 BASIC は気軽にグラフィックスをするのにオススメです。

まず十進 BASIC の説明書チュートリアル (印刷配布してある文書) (<http://www.koshigaya.bunkyo.ac.jp/shiraish/basic/tutorial/contents.htm>) の 2.6 節<sup>3</sup> を読んでみましょう。

### 3.2 チュートリアルを忘れた人向け

チュートリアルのコピーを持って来るのを忘れた (持って来て下さい)、という人のために (コンピューターの画面上で見られますが、自分で打ち込む場合は紙を見ながらの方がやりやすいと思います)、「2.6 グラフィックス」の例題プログラムをここに引用しておきます。

#### 例 14 (関数のグラフ)

```
10 DEF f(x)=x^3-3*x+1
20 SET WINDOW -4,4,-4,4
30 DRAW GRID
40 FOR x=-4 TO 4 STEP 0.1
50   PLOT LINES: x,f(x);
60 NEXT x
70 END
```

(要点: 1 変数実数値関数のグラフ  $y = f(x)$  の描き方の定跡みたいなプログラム)

#### 例 15 (OPTION ANGLE DEGREES, DRAW GRID())

```
10 OPTION ANGLE DEGREES
20 DEF f(x)=sin(x)
30 SET WINDOW -360,360,-4,4
40 DRAW GRID(90,1)
50 FOR x=-360 TO 360
60   PLOT LINES: x,f(x);
70 NEXT x
80 END
```

(要点: 三角関数の単位を度にする OPTION ANGLE DEGREES と DRAW GRID(,) の使い方)

<sup>2</sup>多くの場合、特定のグラフィックス・ライブラリ (例えば GLSC) を呼び出して描画することになり、使用する環境に依存したプログラムとなります。

<sup>3</sup><http://www.koshigaya.bunkyo.ac.jp/shiraish/basic/tutorial/section2.htm\#2.6>

### 例 16 (“PLOT LINES” で線を切る)

2つの関数のグラフを描くために、途中で「切る」必要がある。

```
100 DEF f(x)=x^2
110 DEF g(x)=x^3
120 SET WINDOW -4,4,-4,4
130 DRAW GRID
140 FOR x=-4 TO 4 STEP 0.1
150   PLOT LINES: x,f(x);
160 NEXT x
170 PLOT LINES
180 FOR x=-4 TO 4 STEP 0.1
190   PLOT LINES: x,g(x);
200 NEXT x
210 END
```

(要点: “PLOT LINES: x,y;” は線をつなげていくので、170行目のセミコロンなしの PLOT LINES で線を切る。)

### 例 17 (パラメーター曲線)

```
10 OPTION ANGLE DEGREES
20 DEF f(t)=3*COS(t)
30 DEF g(t)=2*SIN(t)
40 SET WINDOW -4,4,-4,4
50 DRAW grid
60 FOR t=0 TO 360
70   PLOT LINES: f(t),g(t);
80 NEXT t
90 END
```

(要点: パラメーター曲線  $x = f(t)$ ,  $y = g(t)$  の描き方の定跡みたいなプログラム)

### 例 18 (極方程式表示の曲線)

```
10 DEF f(t)=SIN(2*t)
20 SET WINDOW -1,1,-1,1
30 DRAW grid
40 FOR t=0 TO 2*PI STEP PI/360
50 PLOT LINES: f(t)*COS(t), f(t)*SIN(t);
60 NEXT t
70 END
```

(要点: 極方程式  $r = f(\theta)$  は、 $x = f(\theta) \cos \theta$ ,  $y = f(\theta) \sin \theta$  として単なるパラメーター曲線)

### 例 19 (直交座標から極座標への変換)

```
100 SET WINDOW -4,4,-4,4
110 DRAW grid
120 FOR t=0 TO 2*pi STEP pi/180
130   LET x=cos(t)+1
140   LET y=sin(t)
150   LET r=x^2+y^2
160   LET a=ANGLE(x,y)*2
170   PLOT LINES: r*cos(a),r*sin(a);
180 NEXT t
190 END
```

余談 3.1 (直交座標 ↔ 極座標変換) 例 19 は率直に言って、お手本とするには不適切なプログ

ラムだと思えます。  $x = r \cos \theta$ ,  $y = r \sin \theta$  の  $x, y$  (直交座標) から  $r, \theta$  (極座標) を求めるには、

```
R=SQR(X^2+Y^2)
THETA=ANGLE(X,Y)
```

のようにする、と覚えて下さい。

それはさておき、直交座標から極座標に変換するには、C 言語のプログラムだったら、

```
r=sqrt(x*x+y*y); (または r=hypot(x,y);) そして theta=atan2(y,x);
```

とするところです。atan2(,) という得体の知れない名前の関数よりは、ANGLE(,) の方が覚えやすく良い (BASIC の規格の方が趣味が良い)、と私は思います。 ■

メモ: hypotenuse (直角三角形の) 斜辺

### 3.3 まとめ

- `SET WINDOW  $x_1, x_2, y_1, y_2$`  は座標系の設定
- `DRAW GRID` はグリッド (格子、網目という意味) を描く  
`DRAW GRID( $x, y$ )` というようにグリッドの幅も指定できる
- `PLOT LINES:  $x, y$` ; 次に実行される PLOT LINES 命令で指定される点との間を線分で結ぶ。線を区切るには、単に `PLOT LINES` とする。
- 1 変数関数のグラフ、2 次元パラメーター曲線は描くのが簡単 (例題プログラム参照)。
- 他に
  - 点を描く `PLOT POINT:  $x, y$`  命令
  - 文字を描く `PLOT TEXT ,AT  $x, y$ : "文字列"` 命令
  - 多角形領域を塗り潰す `PLOT AREA:  $x_1, y_1; \dots; x_n, y_n$`  命令

などもある<sup>4</sup>。これらの命令の使い方についてはオンライン・ヘルプを参照。

- 色も指定できる。

`SET POINT COLOR  $c$` , `SET LINE COLOR  $c$` , `SET AREA COLOR  $c$` , `SET TEXT COLOR  $c$`  等々。ここで  $c$  は色番号で、以下のように定義されている。

0 白, 1 黒, 2 青, 3 緑, 4 赤, 5 水色, 6 黄色, 7 赤紫, 8 灰色, 9 濃い青, 10 濃い緑,  
11 青緑, 12 えび茶, 13 オリーブ色, 14 濃い紫, 15 銀色, …

<sup>4</sup>area は数学語としては「面積」ですが、「範囲」、「地方」などの意味があるわけですね (知っていると思いますが)。text は「本文」、「教科書」、「文章」。

- グラフのウィンドウ (なんとか .BMP のような名前がついている) の「ファイル」メニューで、「名前をつけて保存 (A)」から描いたグラフィックスをファイルに保存できますが、「ファイルの種類 (T)」で、BitMap でなく、JPEG や GIF フォーマットを選ぶことを勧めます (画像ファイルはサイズが大きくなりがちで、特別の理由がない限り、何らかの方法で圧縮したフォーマットを選ぶべきです,  $\text{\TeX}$  に取り込む場合は JPEG が良いと思います)。

圧縮する場合に、現在使用している Windows の「名前をつけて保存」メニューでは、圧縮率 (≒画像の品質) の設定が出来ないので (そのせいで、ボケた画像になるのがちょっと不満)、一度 BitMap (24ビット) で出力してから、画像処理ソフトを用いて適当な圧縮率で JPEG や GIF に変換する、というやり方は考えられなくもないです。

## 4 レポート課題 6A

十進 BASIC で円を描く<sup>5</sup>プログラム `kadai6a.BAS` と描いた図を含んだ、`kadai6a.pdf` を送って下さい。色はお好みで。

これは出席代りで、なるべく今日 (5月22日) 授業時間中に書き込むこと。

画像を  $\text{\TeX}$  文書に取り込む方法については、「画像の取り込み」<sup>6</sup> を参考にして下さい。今回の場合の手順は以下に説明します。

1. 十進 BASIC のグラフィックスのウィンドウ (プログラムの名前が `kadai6a.BAS` の場合、グラフィックスのウィンドウの名前は最初は `kadai6a.BMP` となっているはず) で [ファイル] メニューの [名前をつけて保存 (A)] で、JPEG 形式で保存する。ここでは `kadai6a.JPG` という名前のファイルになったとする。
2. JPEG 形式を EPS 形式に変換する。情報処理 2 用のフォルダに作った [コマンドプロンプト] を開いて、

```
Z:\¥.windows2000¥syori2>jpeg2ps kadai6a.JPG > kadai6a.eps
```

とコマンドを実行する (`jpeg2ps` は情報処理教室にはインストール済みのはず)。これで `kadai6a.eps` というファイルが出来るはず。

3.  $\text{\TeX}$  文書 (`kadai6a.tex`) のプリアンブル (`\begin{document}` の前) に

```
\usepackage[dvipdfm]{graphicx}
```

と書き、グラフを取り込みたいところで、例えば

<sup>5</sup>昔、パソコンで BASIC 華やかなりし頃は「(楕)円を描く命令」が備わっていました。でも十進 BASIC にはありません。ないけれども“簡単に描ける”はず。とりあえずノーヒントにします。

<sup>6</sup><http://www.math.meiji.ac.jp/~mk/labo/text/tex/node57.html>

center 環境で中央に配置, 横幅 8cm にする

```
\begin{center}
  \includegraphics[width=8cm]{kadai6a.eps}
\end{center}
```

あるいは

figure 環境で図を配置

```
\begin{figure}[htbp]
\centering
\includegraphics[width=8cm]{kadai6a.eps}
\caption{☆を描く}
\end{figure}
```

とする (図を取り込むような場合、後者のように figure 環境を使うのが良いと思いますが、コントロールの難しいところもあって、説明をサボりたいので、center 環境でも構いません)。

もう一回くらい、お手本を見せます (そろそろ .tex を見せるのは止めようかな、と思っ  
ていますが)。

```

kadai6a.tex
\documentclass[12pt]{jarticle}
\usepackage[a4paper]{geometry}% 好みの問題
\usepackage{amsmath,amssymb}% 今回は不要かも
\usepackage{moreverb}% 今回これが必要
\usepackage[dvipdfm]{graphicx}% 今回これが必要
\begin{document}
\title{情報処理2 課題6A レポート}
\author{2年16組99番 数学 学}
\date{2013年5月22日}
\maketitle

\section{プログラム}

\verbatiminput{kadai6a.BAS}

\section{プログラムの実行結果}

kadai6a.BAS の実行結果は次のようになる。

\begin{center}
\includegraphics[width=8cm]{kadai6a.eps}
\end{center}

\end{document}

```

## 5 レポート課題6B

(多分、5/22 はここまで出来ないと思います。)

実際、全然たどり着きませんでした。内容は第7回<sup>7</sup>に引っ越します。

---

<sup>7</sup><http://www.math.meiji.ac.jp/~mk/syori2-2013/jouhousyori2-2013-07/>